



**ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ**

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: [ogu.ahmet.topcu@gmail.com](mailto:ogu.ahmet.topcu@gmail.com)

Web: <http://mmf2.ogu.edu.tr/atopcu>

**Bilgisayar Destekli**

# **Nümerik Analiz**

*Ders notları 2014*

**Ahmet TOPÇU**

```
C:\Basic\QBasic.EXE
File Edit View Search Run Debug Calls Utility Options Help
GAUSSSEI.BAS:GaussSeidel
DEFINT I-N
DEFDBL A-H, O-Z
SUB GaussSeidel (n, a(), b(), x(), itersay, ihata)
'-----
' Doğrusal denklem sistemi çözümü (GaussSeidel)
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' A(n,n)*x(n)=b(n) denklem GAUSS-SEIDEL iterasyon
' yöntemi ile çözümlenerek x(n) vektörü hesaplanır.
' A, b matrisleri çağırılan programda depolanmış ve x
' vektörü boyutlandırılmış olmalıdır.
' Maxit: öngörülen maksimum iterasyon sayısıdır
' Eps: hassasiyet
' ihata=0: çözüm yakınsadı
' ihata=1: diyagonal eleman sıfır, matris tekil
' ihata=2: Max iterasyon sayısı aşıldı
' ihata=3: istenilen hassasiyete varılamadı
' ihata=4: çözüm iraksıyor, sayı taşması tehlikesi!
'-----
' EpsMach
EpsMach = 1
DO
  EpsMach = EpsMach / 2
  s = 1 + EpsMach
LOOP UNTIL s <= 1
EpsMach = 2 * EpsMach

  ihata = 0
  maxit = 200: ' maximum iterasyon sayısı
  Eps = .000001: 'tolerans
  itersay = 0: ' iterasyon no

' Pivot eleman ara, diyagonal elemanı<>0 yap
FOR i = 1 TO n
  T = ABS(a(i, i)): iU = i: x(i) = 0
  FOR j = i + 1 TO n
    IF ABS(a(j, i)) > T THEN iU = j: T = ABS(a(j, i))
  NEXT j

  IF T < EpsMach THEN ihata = 1: EXIT SUB
  IF iU > i THEN
```

# 21

**PROGRAMLAR:**

**İterasyon ile denklem sistemi çözümü  
Gauss-Seidel metodu**

## 21. PROGRAMLAR: İterasyon ile denklem sistemi çözümü-GAUSS-SEIDEL iterasyon yöntemi<sup>1</sup>

Ekte verilen program  $\underline{A} \underline{x}=\underline{b}$  lineer cebirsel denklem sistemini GAUSS-SEIDEL iterasyon yöntemi ile çözer.  $\underline{A}_{n \times n}$  katsayılar matrisi ve  $\underline{b}_n$  karşı taraf vektörü çağırın programda depolanmış olmalı ve ayrıca  $\underline{x}_n$  vektörü de boyutlandırılmış olmalıdır.

GAUSS-SEIDEL yöntemi  $\underline{A}_{n \times n}$  kesin diyagonal ağırlıklı ise veya simetrik ve pozitif tanımlı ise daima yakınsar (sonuç verir). Diğer matris tiplerinde çözüm yakınsamayabilir.

Yöntemin gereği olarak  $\underline{A}$  nın diyagonal elemanları sıfırdan farklı olmalıdır. Bu nedenle program  $\underline{A}$  nın diyagonal elemanları mutlak değerce en büyük olacak şekilde denklemlerin sırasını çözüm öncesi değiştirir. Tüm diyagonal elemanlar sıfırdan farklı yapılamazsa program hesabı yarıda keser ve çağırın programa geri döner.

İterasyon sayısı Eps istenilen hassasiyetine bağlıdır. İterasyon sayısı Maxit aşılırsa veya Eps hassasiyetine ulaşamazsa veya çözüm iraksarsa hesap kesilir, iHata $\neq$ 0 değeri ile döner.

$\underline{A}$  nın kesin diyagonal ağırlıklı<sup>2</sup> olmaması veya simetrik ve pozitif tanımlı olmaması durumunda çözümün yakınsayacağı garanti edilemez, bu durumda GAUSS-SEIDEL yöntemi önerilmez. Çok büyük denklem sistemlerin GAUSS veya CHOLESKY direkt yöntemleri ile çözümü sınırlı bellek açısından sorun yaratırsa veya  $\underline{A}$  seyrek (sparse) matris ise GAUSS-SEIDEL tercih edilebilir.

### Örnek 1:

$$\begin{bmatrix} 10 & 2 & 3 & -1 \\ 1 & -20 & -1 & 3 \\ 1 & 1 & -10 & 2 \\ 2 & -1 & -1 & 30 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 80 \\ 40 \\ 40 \\ 120 \end{bmatrix} \Rightarrow \text{(kesin Diyagonal ağırlıklı)}$$

**GaussSeidel alt programının sonucu**

```
C:\Basic\QBASIC.EXE
İterasyon sayısı= 8
Denklem sisteminin çözümü (GaussSeidel):
9.2646045653036 -0.920978240674928 -2.51205223229254 3.26792534654751
```

### Örnek 2:

$$\begin{bmatrix} 10 & 2 & 3 & -1 \\ 1 & -20 & -1 & 3 \\ 1 & 1 & -1 & 2 \\ 2 & -1 & -1 & 1 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 80 \\ 40 \\ 40 \\ 120 \end{bmatrix} \Rightarrow \text{(Kesin diyagonal ağırlıklı değil. Çözüm var fakat iterasyon yöntemi çözümüne güvenilmez!)}$$

**Çözüm var, fakat GaussSeidel iraksıyor**

```
C:\Microsoft QuickBASIC
İterasyon sayısı= 91
Çözüm iraksıyor!
```

<sup>1</sup> Teori ve sayısal örnekler için bak: bölüm 7

<sup>2</sup> Bak: Bölüm 1, sayfa 15

```

'----- ana program GAUSS-SIDEL-----
' iterasyon yöntemi ile denklem sistemi çözümü
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' Çağrılan alt programlar: GaussSeidel
'-----
DATA 4: ' denklem sayısı
'a matrisi:
DATA 10, 2, 3, -1
DATA 1, -20,-1, 3
DATA 1, 1,-10, 2
DATA 2, -1,-1, 30
'b matrisi:
DATA 80,40,40,120

DEFINT I-N
DEFDBL A-H, O-Z
DECLARE SUB GaussSeidel (n, a(), b(), x(), itersay, ihata)

CLS
READ n
DIM a(n, n), b(n), x(n)
FOR i = 1 TO n
  FOR j = 1 TO n
    READ a(i, j)
  NEXT j
NEXT i

FOR i = 1 TO n
  READ b(i)
NEXT i
CALL GaussSeidel(n, a(), b(), x(), itersay, ihata)

PRINT "İterasyon sayısı="; itersay

SELECT CASE ihata
CASE 0
  PRINT "Denklem sisteminin çözümü (GaussSeidel):"
  FOR i = 1 TO n
    PRINT x(i);
  NEXT i
CASE 1
  PRINT "Matris tekil!"
CASE 2
  PRINT "Max itersasyon sayısı aşıldı!"
CASE 3
  PRINT "İstenilen hassasiyete varılamadı!"
CASE 4
  PRINT "Çözüm iraksıyor!"
END SELECT

END 'Gauss-Seidel ana sonu

```

### GaussSeidel Alt programı

```

SUB GaussSeidel (n, a(), b(), x(), itersay, ihata)
'-----
' Doğrusal denklem sistemi çözümü (GaussSeidel)
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
'  $A(n,n) \cdot x(n) = b(n)$  denklem GAUSS-SEIDEL iterasyon
' yöntemi ile çözülerek  $x(n)$  vektörü hesaplanır.
' A, b matrisleri çağırılan programda depolanmış ve x
' vektörü boyutlandırılmış olmalıdır.
' Maxit: öngörülen maksimum iterasyon sayısıdır
' Eps: hassasiyet
' ihata=0: çözüm yakınsadı
' ihata=1: diyagonal eleman sıfır, matris tekil
' ihata=2: Max iterasyon sayısı aşıldı
' ihata=3: istenilen hassasiyete varılamadı
' ihata=4: çözüm iraksıyor, sayı taşması tehlikesi!
'-----
' EpsMach
EpsMach = 1
DO
  EpsMach = EpsMach / 2
  s = 1 + EpsMach
LOOP UNTIL s <= 1
EpsMach = 2 * EpsMach

  ihata = 0
  maxit = 200: ' maximum iterasyon sayısı
  Eps = .000001: ' tolerans
  itersay = 0: ' iterasyon no

' Pivot eleman ara, diyagonal elemanı <>0 yap
FOR i = 1 TO n
  T = ABS(a(i, i)): iV = i: x(i) = 0
  FOR j = i + 1 TO n
    IF ABS(a(j, i)) > T THEN iV = j: T = ABS(a(j, i))
  NEXT j

  IF T < EpsMach THEN ihata = 1: EXIT SUB
  IF iV > i THEN

' i ve iV satırlarını deęiş
  FOR j = 1 TO n
    T = a(i, j)
    a(i, j) = a(iV, j)
    a(iV, j) = T
  NEXT j
  T = b(i)
  b(i) = b(iV)
  b(iV) = T
END IF
NEXT i

' İterasyon
DO
  h = 0
  FOR i = 1 TO n
    T = 0
    FOR j = 1 TO n
      T = T + a(i, j) * x(j)
    NEXT j
    T = T - a(i, i) * x(i)
    YeniX = (b(i) - T) / a(i, i)
    h = (x(i) - YeniX) ^ 2
    x(i) = YeniX
  NEXT i
  h = SQR(h)
  itersay = itersay + 1

LOOP UNTIL (itersay > maxit OR h <= Eps OR h > 1E+35)
IF itersay > maxit THEN ihata = 2
IF h > Eps THEN ihata = 3
IF h > 1E+35 THEN ihata = 4

END SUB ' GaussSeidel sonu

```